

Problems in Mathematics & Experiments with Mathematica

6. Definite integral

6.3 Numerical integration: Integral approximation formulas

Theory

Newton-Leibniz formula cannot be applied if there is no antiderivative (at least piecewise defined), or the function is given by experimental data. In such cases numerical integration formulas are used. For exercises, we consider here only the simplest formulas. Advanced methods resulting higher precision are implemented in the *Mathematica* command `NIntegrate`.

Let $f(x)$ be integrable on the interval $[a, b]$. Calculate approximately $\int_a^b f(x) dx$. Let n be a big integer, and $x_i = a + i(b - a)/n$, $i = 0, 1, \dots, n$. Approximate $f(x)$ on $[x_i, x_{i+1}]$ by $f(x_i)$ or $f(x_{i+1})$. Then we obtain the approximate values

$$I_{\text{left}} = \frac{b-a}{n} \sum_{i=0}^{n-1} f(x_i)$$

and

$$I_{\text{right}} = \frac{b-a}{n} \sum_{i=0}^{n-1} f(x_{i+1}),$$

respectively. Averaging them, we obtain the trapezoidal formula

$$I_{\text{trapez}} = \frac{I_{\text{left}} + I_{\text{right}}}{2} = \frac{b-a}{n} \left(\frac{f(a) + f(b)}{2} + \sum_{i=1}^{n-1} f(x_i) \right)$$

that approximates the integral on $[x_i, x_{i+1}]$ by the straight line $y = f(x_i) + \frac{f(x_{i+1}) - f(x_i)}{x_{i+1} - x_i} (x - x_i)$.

If the function is given by experimental data, that is, the data set is $(x_i, f(x_i))$, $i = 0, 1, 2, \dots, n$, then the function can be considered as step-function (for left and right formulas), or a piecewise linear function for the trapezoidal formula. The given sequence $x_0, x_1, x_2, \dots, x_n$ (which is assumed to be increasing) has to be used in the

above formulas instead of any uniform grid. In this case, the trapezoidal formula has the form

$$I_{\text{trapez}} = \left(\sum_{i=0}^{n-1} (x_{i+1} - x_i) \frac{f(x_i) + f(x_{i+1})}{2} \right)$$

Important! If some of the distances $x_{i+1} - x_i$ are not small enough then the piecewise linear approximation can be very far from the reality.

Mathematica statements

Mathematica	Meaning
<code>NIntegrate[f[x], {x, a, b}]</code>	The definite integral of $f(x)$ on $[a, b]$

For more details, see the *Mathematica* Help on [NIntegrate](#).

Exercises and problems

■ Mathematica initialization

```
LeftSum[f_, {x_, x0_, x1_}, nn_] :=
  ((x1 - x0)*Sum[f, {x, x0, x1 - (x1 - x0)/nn,
    (x1 - x0)/nn}])/nn

RightSum[f_, {x_, x0_, x1_}, nn_] :=
  ((x1 - x0)*Sum[f, {x, x0 + (x1 - x0)/nn, x1,
    (x1 - x0)/nn}])/nn

TrapezoidSum[f_, {x_, x0_, x1_}, nn_] :=
  1/nn*(x1 - x0)*(1/2*
    ((f /. x -> x0) + (f /. x -> x1)) +
    Sum[f, {x, x0 + (x1 - x0)/nn, x1 - (x1 - x0)/nn,
      (x1 - x0)/nn}])

ListTrapezoidSum[f1_] :=
  Sum[1/2*(f1[[i,2]] + f1[[i + 1,2]])*
    (f1[[i + 1,1]] - f1[[i,1]]),
    {i, 1, Length[f1] - 1}]
```

SOLVED PROBLEM 6.3.1

Use the left, right and trapezoidal formulas to approximate the following integral

$$\int_1^3 x^2 dx$$

Verify your calculation by the function `LeftSum`, `RightSum`, and `TrapezoidSum`, respectively. Increase n to improve the accuracy. Use `NIntegrate` to compare the accuracy of the trapezoidal formula and the built-in *Mathematica* function.

◦ **SOLUTION**

Use `LeftSum`, `RightSum`, and `TrapezoidSum`:

```
x0 = 1; x1 = 3; n = 10;
```

```
LeftSum[x^2, {x, x0, x1}, n]
```

$$\frac{197}{25}$$

```
RightSum[x^2, {x, x0, x1}, n]
```

$$\frac{237}{25}$$

```
TrapezoidSum[x^2, {x, x0, x1}, n]
```

$$\frac{217}{25}$$

Now, use `Integrate` and `NIntegrate`:

$$\int_{x_0}^{x_1} x^2 dx$$

$$\frac{26}{3}$$

```
NIntegrate[x^2, {x, x0, x1}]
```

```
8.66667
```

◦

PROBLEM 6.3.2

Use the trapezoidal formula and verify your calculations with *Mathematica*.

(1) $n = 5; \int_0^1 \sqrt{x-2} dx$

(2) $n = 4; \int_{-1}^1 |x^3| dx$

(3) $n = 5; \int_1^{10} \frac{1}{x} dx$

(4) $n = 3; \int_0^\pi \sin(x) dx$

SOLVED PROBLEM 6.3.3 Integrate a function given by data

A sample of the function $f(x)$ is given in the following list. Assume that the elements $(x_i, f(x_i))$ are sorted by x_i . Use the trapezoidal formula to calculate the integral on $[x_0, x_n]$.

$$\text{flist} = \begin{pmatrix} 0.1871420 & 0.0350220 \\ 0.7340650 & 0.5388510 \\ 0.2153529 & 0.0463769 \\ 0.2045849 & 0.0418552 \\ 0.7799850 & 0.6083769 \\ 0.0086980 & 0.0000756 \\ 0.8916940 & 0.7951179 \\ 0.9145049 & 0.8363189 \\ 0.2094300 & 0.0438609 \\ 0.8769579 & 0.7690549 \end{pmatrix};$$

◦ SOLUTION


The list was generated by

```
flist = Table[{#, #^2} & [Random[]], {i, 1, 10}];
```

Note that the Monte-Carlo numeric integration methods are based on such random data generations.

- *Plot the data:*

```
ListPlot[flist];
```



- *The list is not sorted. Use the function ListTrapezoidSum for the sorted list.*

```
ListTrapezoidSum[Sort[flist]]
```

```
0.279319
```

◦

PROBLEM 6.3.4

Plot and integrate the functions given by the following data.

$$(1) \text{ flist} = \begin{pmatrix} 1. & 0.54 \\ 2. & -0.416 \\ 3. & -0.989 \\ 4. & -0.653 \\ 5. & 0.2836 \end{pmatrix};$$

$$(2) \text{ flist} = \begin{pmatrix} 1 & 1 \\ 2 & 2\sqrt{2} \\ 3 & 3\sqrt{3} \\ 4 & 8 \\ 5 & 5\sqrt{5} \end{pmatrix};$$

$$(3) \text{ flist} = \begin{pmatrix} 1. & 0 \\ 2. & 0.6931 \\ 3. & 1.0986 \\ 4. & 1.3862 \\ 5. & 1.6094 \end{pmatrix};$$

$$(4) \text{ flist} = \begin{pmatrix} 1 & 0.5 \\ 2 & 0.25 \\ 3 & 0.125 \\ 4 & 0.0625 \\ 5 & 0.03125 \end{pmatrix};$$

$$(5) \text{ flist} = \begin{pmatrix} 2.1027 & 0.86181 \\ 0.1566 & 0.15602 \\ 1.3831 & 0.98244 \\ 2.0133 & 0.90366 \\ 2.0943 & 0.86606 \\ 3.0889 & 0.05261 \\ 2.6482 & 0.47353 \\ 0.1488 & 0.14826 \\ 0.9872 & 0.83449 \\ 0.9761 & 0.82834 \end{pmatrix};$$

$$(6) \text{ flist} = \begin{pmatrix} 1.919 & 1.385 \\ 1.306 & 1.142 \\ 0.082 & 0.286 \\ 0.462 & 0.680 \\ 1.683 & 1.297 \\ 0.458 & 0.676 \\ 1.523 & 1.234 \\ 1.072 & 1.035 \\ 1.505 & 1.226 \\ 0.970 & 0.985 \end{pmatrix};$$
